
Abstract. Funnel plots are commonly used to investigate publication and related
biases in meta-analysis. Although asymmetry in the appearance of a funnel plot
is often interpreted as being caused by publication bias, in reality the asymmetry
could be due to other factors that cause systematic differences in the results of
large and small studies, for example, confounding factors such as differential study
quality. Funnel plots can be enhanced by adding contours of statistical significance
to aid in interpreting the funnel plot. If studies appear to be missing in areas of low
statistical significance, then it is possible that the asymmetry is due to publication
bias. If studies appear to be missing in areas of high statistical significance, then
publication bias is a less likely cause of the funnel asymmetry. It is proposed
that this enhancement to funnel plots should be used routinely for meta-analyses
where it is possible that results could be suppressed on the basis of their statistical
significance.

1 Introduction

Publication bias is the phenomenon where studies with uninteresting or unfavorable
results are less likely to be published than those with more favorable results (Rothstein,
Sutton, and Borenstein 2005). If publication bias exists, then the published literature
is a biased sample of all studies, and any meta-analysis based on it will be similarly
biased.

Funnel plots are commonly used to investigate publication and related biases in
meta-analysis (Sterne, Becker, and Egger 2005). They consist of a simple scatterplot of
each study’s estimate of effect against some measure of its variability, commonly plotted
on the x and y axes, respectively (although this goes against the usual convention of
plotting the response variable on the y axis). In this way, the studies with the least
variable effect sizes appear at the top of the funnel, and the smaller, less precise studies
appear at the bottom. In the absence of publication bias, the studies will fan out in

a symmetrical funnel shape around the pooled estimate, as variability due to sampling
error increases down the y axis. If publication bias is present, then the funnel will
appear asymmetric because of the systematic suppression of studies.

A complication in interpreting funnel plots is that funnel asymmetry could be due
to factors other than publication bias, such as systematic differences in the results of
large and small studies caused by confounding factors such as differential study quality;
these differences are sometimes called small-study effects (Sterne and Egger 2001). The
aim of the contour-enhanced funnel plot is to aid in disentangling these different causes
of funnel asymmetry (Peters et al. 2008).

Funnel plots in Stata were previously described by Sterne and Harbord (2004),
and there are several commands available in Stata for drawing funnel plots includ-
ing metafunnel, funnel (available with metan), and metabias. These commands are
described in more detail in a frequently asked question about the Stata commands avail-
able for meta-analysis; the frequently asked question can be found on Stata’s web site
at http://www.stata.com/support/faqs/stat/meta.html. In Stata 10, typing help meta

displays a help file with information about the user-written commands for meta-analysis
and tells which are the latest versions.

This article introduces another command for meta-analysis called confunnel, which
produces contour-enhanced funnel plots. The concept of the contour-enhanced funnel
plot is explained in the next section, followed by a description of the command syntax
and options. The use of confunnel is demonstrated on a well-known meta-analysis
example, and the use of the command is also explained in conjunction with some of the
other user-written meta-analysis commands.

2 Contour-enhanced funnel plots

There is evidence that, generally, the primary driver for the suppression of studies is
the level of statistical significance of study results, with studies that do not attain
perceived milestones of statistical significance (i.e., p < 0.05 or 0.01) being less likely
to be published (Easterbrook et al. 1991; Dickersin 1997; Ioannidis 1998). Despite this,
no method has been previously considered to identify the areas of the funnel plot that
correspond to different levels of statistical significance, to assess whether any observed
asymmetry is likely caused by publication bias.

On a contour-enhanced funnel plot, contours of statistical significance are overlaid
on the funnel plot (Peters et al. 2008). Adding contours of statistical significance in
this way facilitates the assessment of whether the areas where studies exist are areas
of statistical significance and whether the areas where studies are potentially missing
correspond to areas of low statistical significance. If studies appear to be missing in
areas of low statistical significance, then it is possible that the asymmetry is due to
publication bias. Conversely, if the area where studies are perceived to be missing are
areas of high statistical significance, then publication bias is a less likely cause of the
funnel asymmetry.

There has been discussion as to which is the most informative scale for funnel plots
of binary outcome meta-analyses. The consensus is that using the standard error, the
variance, or their inverses is most sensible over using an alternative such as sample size
(Sterne and Egger 2001; Sterne, Becker, and Egger 2005). Using the standard error on
the y axis is easiest to interpret because, in this instance, the contours of statistical
significance are linear, which is because they are derived from the Wald statistic for
each study’s effect estimate. The confunnel command has an option to use standard
error, inverse standard error, variance, or inverse variance on the y axis.

A meta-analysis of trials investigating magnesium therapy following myocardial in-
farction is a well-known example in the literature where the presence of publication
bias is suspected (Teo et al. 1991; ISIS-4 Collaborative Group 1995; Sterne, Bradburn,
and Egger 2001). An initial meta-analysis found that magnesium therapy reduced the
risk of mortality; however, a number of larger trials were subsequently published that
found no evidence that magnesium therapy reduced the risk of mortality. A standard
funnel plot is given for this meta-analysis in figure 1, which was generated by using the
metafunnel command as shown in the following syntax:

. use magnesium

. gen logES = logor

. gen selogES = selogor

. metafunnel logES selogES

Figure 1. metafunnel funnel plot

When the standard error is used on the y axis of a funnel plot, it is conventional to
reverse the axis so that the most precise studies are displayed at the top of the plot.

Figure 1 is compared with the equivalent funnel plot produced by confunnel, shown
in figure 2. The addition of the contours of statistical significance makes it easier to

assess the proportion of studies published in the meta-analysis at and around statistical
significance. The syntax for the default confunnel plot, with the sj scheme, is

. confunnel logES selogES

Figure 2. confunnel funnel plot using default options

In both figures 1 and 2, there is a strong suggestion of asymmetry in the funnel,
suggesting that studies are missing on the right-hand side of the plot, but figure 2
makes it easier to assess the statistical significance of the hypothetical missing studies.
The area where missing studies are perceived includes regions of both low and high
statistical significance (i.e., the area crosses over the contours), suggesting studies that
showed magnesium to be nonsignificantly and significantly less effective to be missing.
Therefore, publication bias cannot be accepted as the only cause of funnel asymmetry
if it is believed studies are being suppressed because of a mechanism based on two-sided
p-values.

It is important to emphasize the differences between the pseudo 95% confidence
limits produced by metafunnel on figure 1 and the contours of statistical significance
produced by confunnel on figure 2 (Peters et al. 2008). The pseudo 95% confidence
limits illustrate the expected 95% confidence interval about the pooled fixed-effects
estimate for the meta-analysis. The pseudo-confidence limits therefore help to assess the
extent of between-study heterogeneity in the meta-analysis and the asymmetry on the
funnel plot. Unlike the pseudo-confidence limits, the contours of statistical significance
are independent of the pooled estimate; therefore, if the pooled estimate is subject
to bias, then the contours of significance will not be affected. Also, when the pooled
estimate is at the null, the pseudo 95% confidence limits coincide with the two-sided
5% significance contours.

3 The confunnel command

The confunnel command plots contour-enhanced funnel plots for study outcome mea-
sures in a meta-analysis. Contours of statistical significance from one- or two-sided
Wald tests can be plotted using shaded or dashed contour lines. Contours can be plot-
ted along any number of chosen levels of statistical significance; by default, 1%, 5%,
and 10% significance contours are plotted. As previously mentioned, confunnel has
the choice of four y axes. The command also has been designed to be flexible, allowing
the user to add extra features to the funnel plot.

3.1 Syntax

confunnel varname1 varname2
[
if
] [

in
] [

, aspectratio(string)

contours(numlist) contcolor(color) extraplot(plots)

functionlowopts(options) functionuppopts(options) legendlabels(labels)

legendopts(options) metric(se | invse | var | invvar) onesided(lower | upper)
scatteropts(options) shadedcontours solidcontours twowayopts(options)

]

The first variable, varname1 is the variable corresponding to the effect estimates, often
log odds ratios, and the second variable, varname2, is the variable corresponding to the
standard errors of the effect estimates.

3.2 Options

aspectratio(string) specifies the aspect ratio for the plot; the default is 1.

contours(numlist) specifies the significance levels of the contours to be plotted; the
default is set to 1%, 5%, and 10% significance levels.

contcolor(color) specifies the color of the contour lines if shadedcontours is not
specified.

extraplot(plots) specifies one or multiple additional plots to be overlaid on the funnel
plot.

functionlowopts(options) and functionuppopts(options) pass options to the twoway
function commands used to draw the significance contours; for example, the line
widths can be changed.

legendlabels(labels) specifies labels to appear in the legend for extra elements added
to the funnel plot.

legendopts(options) passes options to the plot legend.

metric(se | invse | var | invvar) specifies the metric of the y axis of the plot. se,
invse, var, and invvar stand for standard error, inverse standard error, variance,
and inverse variance, respectively; the default is se.

onesided(lower | upper) can be lower or upper, for lower-tailed or upper-tailed levels
of statistical significance, respectively. If unspecified, two-sided significance levels
are used to plot the contours.

scatteropts(options) specifies any of the options documented in [G] graph twoway

scatter.

shadedcontours specifies shaded, instead of black, contour lines.

solidcontours specifies solid, instead of dashed, contour lines.

twowayopts(options) specifies options passed to the twoway plotting function.

4 Use of confunnel

The following subsections use the meta-analysis of magnesium therapy following my-
ocardial infarction.

4.1 Demonstration of some confunnel options

Figure 3 shows the use of the inverse standard error on the y axis; the syntax is as
follows:

. confunnel logES selogES, metric(invse)

Figure 3. confunnel funnel plot using inverse standard error on the y axis

If there is strong evidence that studies are suppressed based on a one-sided (rather
than a two-sided) significance test, this can be investigated using the onesided() option,
as shown in figure 4 and in the following syntax:

. confunnel logES selogES, onesided(lower)

Figure 4. confunnel using lower tail one-sided significance regions

Unlike figure 2, in figure 4 (based on one-sided p-values) the area where studies
are perceived missing is within the region of low statistical significance. Under this
assumption, it is more reasonable to consider publication bias as the potential cause of
the funnel asymmetry. In this context, the one-sided assumption implies that studies
showing magnesium to be harmful are likely to be suppressed regardless of the signifi-
cance of the results. Previous methods to address publication bias have made various
assumptions about the sidedness of suppression; for example, the trim-and-fill method
is one-sided, whereas Egger’s regression test is two-sided (Duval and Tweedie 2000;
Egger et al. 1997).

Figure 5 shows using variance on the y axis, using the shaded and solid contours
options, and labeling the x axis with odds ratios on the funnel plot. The syntax is
shown here (confunnel was run prior to these commands in order to see where Stata
placed the tick marks on the x axis):

. local t1 = round(exp(-4)*100)/100

. local t2 = round(exp(-2)*100)/100

. local t3 = exp(0)

. local t4 = round(exp(2)*100)/100

. local t5 = round(exp(4)*100)/100

. confunnel logES selogES, metric(var) shadedcontours solidcontours
> twowayopts(xtitle("Odds ratios")
> `"xlabel(-4 "`t1´" -2 "`t2´" 0 "`t3´" 2 "`t4´" 4 "`t5´")"´)

Figure 5. confunnel using variance on the y axis

4.2 Use of confunnel with metan, metabias, metamodbias,
and metatrim

The metan command for meta-analysis (Bradburn, Deeks, and Altman 1998; Harris et
al. 2008) can be used to generate the information to display the pooled fixed-effects
estimate with its pseudo 95% confidence interval (or, indeed, the pooled random-effects
estimate) on the confunnel plot; this is shown in figure 6. In this example, because
the pooled log odds ratio was very close to 0, the pseudo 95% confidence interval (for
the pooled fixed-effects estimate) almost coincided with the 5% significance contours,
which are symmetric about the null hypothesis. The syntax for figure 6 is as follows:

. capture drop logES selogES

. metan alive0 dead0 alive1 dead1, or nograph fixed

(output omitted )

. local fixedlogES = log(r(ES))

. generate logES = log(_ES)

. rename _selogES selogES

. summarize selogES, meanonly

. local semax = r(max)

. confunnel logES selogES, extraplot(function `fixedlogES´, horizontal
> lc(gs8) range(0 `semax´) || function `fixedlogES´ + x*invnormal(.025),
> horizontal range(0 `semax´) lc(gs8) || function `fixedlogES´ +
> x*invnormal(.975), horizontal range(0 `semax´) lc(gs8))
> legendlabels(`"8 "F.E. & 95% C.I.""´) contcolor(gs10)

Figure 6. confunnel with metafunnel features using metan

Egger’s test investigating possible small-study reporting bias can be represented on
the funnel plot by using the information from the metabias command (Egger et al.
1997; Steichen 1998); this is shown in figure 7 and in the following syntax:

. metabias logES selogES, graph(egger)

(output omitted )

. matrix b = e(b)

. local bias = b[1,2]

. local slope = b[1,1]

. summarize selogES, meanonly

. local semax = r(max)

. metamodbias alive0 dead0 alive1 dead1, graph

(output omitted )

. matrix c = e(b)

. local modbias = c[1,2]

. local modslope = c[1,1]

. confunnel logES selogES, contours(5 10) extraplot(function (`bias´*x + `slope´),
> horizontal range(0 `semax´) lc(gs8) || function (`modbias´*x + `modslope´),
> horizontal range(0 `semax´) lc(gs4)) legendlabels(`"6 "Egger" 7 "Harbord""´)

Also shown on the figure is the modified Egger test using the metamodbias command
(Harbord 2003) because Egger’s test has been shown to be biased for binary outcome
meta-analyses (Harbord, Egger, and Sterne 2006). To download the metamodbias com-
mand from within Stata, type net from http://www.epi.bris.ac.uk/user/rogerh/.

The modified Egger’s test is performed on different scales from those of the axes
of the funnel plot, but when all trials have a reasonable sample size with small effect
estimates, it is not unreasonable to view it on a funnel plot.

Figure 7. confunnel with Egger’s and Harbord’s regression tests using metabias and
metamodbias

Applying the trim-and-fill method to this meta-analysis by using the metatrim com-
mand (Steichen 2000) surprisingly resulted in no studies needing to be filled (Duval
and Tweedie 2000). In order to demonstrate confunnel displaying filled studies, a
meta-analysis of the risk of lung cancer from passive smoking is used (Hackshaw, Law,
and Wald 1997; Rothstein, Sutton, and Borenstein 2005). Applying the trim-and-fill
method, the passive smoking meta-analysis produces seven filled studies, shown in fig-
ure 8 and described with the following syntax:

. use passivesmoking, clear

. local n = _N

. metan logOR selogOR, nograph

(output omitted )

. local ES = r(ES)

. summarize selogOR, meanonly

. local semax = r(max)

. metatrim logOR selogOR, save(metatrimdata, replace)

(output omitted )

. use metatrimdata, clear

. local nfilled = _N - `n´

. metan filled fillse, nograph

(output omitted )

. local filledES = r(ES)

. confunnel filled fillse if _n > `nfilled´, contours(5 10) contcolor(gs10)
> extraplot(scatter fillse filled if _n <= `nfilled´, m(T) mc(gs8) ||
> function `ES´, horizontal lc(black) range(0 `semax´) || function `filledES´,
> horizontal lc(gs8) range(0 `semax´))
> legendlabels(`"6 "Filled" 7 "F.E." 8 "F.E. filled""´)

Figure 8. confunnel with filled studies from metatrim. The vertical dotted line shows
the pooled log odds ratio on the original meta-analysis, while the vertical short dash–
dotted line shows the pooled estimate including the filled studies.

It is possible to consider the studies filled by trim and fill as a guide to the likely
location of missing studies. With the contours added to the funnel plot containing the
filled studies, it is possible to assess the projected significance of the missing studies to
determine if it is reasonable to assume such studies could be suppressed by publication
bias based on a p-value selection mechanism. In figure 8, trim and fill estimates that
seven studies are missing, all of which indicate those exposed to passive smoking are at
a reduced risk of lung cancer and all of which are in the region of p > 0.10. Hence, it
is plausible that publication bias is the cause of the observed asymmetry in this funnel
plot.

5 Discussion

The use of the contour-enhanced funnel plot, implemented with the confunnel com-
mand, is recommended to investigate meta-analyses where it is possible that results
could be suppressed on the basis of their statistical significance. In practice, it is sus-
pected that this could include the majority of contexts in which meta-analysis is con-
ducted, certainly in medicine and related disciplines. Exceptions do exist, for example,
where noncomparative effect sizes are combined (e.g., in a surgical case series or for
incidence or prevalence data); statistical significance will often have no meaning, and in
such cases the contours would not be relevant.

An issue with the interpretation of the contour-enhanced funnel plot is that the
significance contours can draw the analyst into thinking that the studies should be

symmetric about the null hypothesis of the Wald test, because this is the point at
which the contours meet when standard error or variance is used on the y axis. But this
should be avoided because the studies should form a symmetric funnel shape centered
around the true underlying effect size and not the null. Because of this, it can be
helpful to plot the meta-analysis pooled estimate for the data on the funnel, although
the analyst should be aware that this too may be biased if publication bias is present.

In conclusion, funnel plots are a useful tool in the assessment of systematic differences
between the effects in smaller and larger studies in a meta-analysis, regardless of the
underlying reason for the differences. Funnel plots can be enhanced by the inclusion
of contours of statistical significance, which aid in the interpretation of whether such
differences in study estimates in a meta-analysis are most likely to be due to publication
bias or other factors.

